AD-A134 870 A MULTIPLE LEG TMA (TARGET MOTION ANALYSIS) PROCEDURE 1/1
WITH PROGRAMS FOR T..(U) NAVAL POSTGRADUATE SCHOOL
MONTEREY CA R H SHUDDE SEP 83 NPS55-83-025

UNCLASSIFIED F/G 9/2 NL

END
DATE
FILMED
12-83
DTIC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

NPS55-83-025

NAVAL POSTGRADUATE SCHOOL

Monterey, California

A MULTIPLE LEG TMA PROCEDURE WITH PROGRAMS
FOR THE HEWLETT-PACKARD HP-41CV,
THE HEWLETT-PACKARD HP-75C,
THE SHARP PC-1500 (TRS-80 PC-2) AND
THE RADIO SHACK TRS-80 MODEL 100
PORTABLE COMPUTERS

by

Rex H. Shudde

September 1983

Approved for public release; distribution unlimited

Prepared for:
Chief of Naval Research
Arlington, VA 22217

NAVAL POSTGRADUATE SCHOOL
Monterey, California

Rear Admiral J. J. Ekelund
Superintendent

David A. Schrady
Provost

This work was supported by the Office of Naval Research, Fleet Activity Support Division, Code 230.

This report was prepared by:

REX H. SHUDDE, Associate Professor
Department of Operations Research

Reviewed by:

ALAN R. WASHBURN, Chairman
Department of Operations Research

Released by:

for WILLIAM M. TOLLES
Dean of Research

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER NPS55-83-025	2. GOVT ACCESSION NO. AD-A134870	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) A MULTIPLE LEG TMA PROCEDURE WITH PROGRAMS FOR THE HEWLETT-PACKARD HP-41CV, THE HEWLETT-PACKARD HP-75C, THE SHARP PC-1500 (TRS-80 PC-2) AND THE RADIO SHACK TRS-80 MODEL 100 PORTABLE COMPUTERS		5. TYPE OF REPORT & PERIOD COVERED Technical
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Rex H. Shudde		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Naval Postgraduate School Monterey, CA 93943		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS
11. CONTROLLING OFFICE NAME AND ADDRESS Office of Naval Research Fleet Activity Support Division, Code 230 Arlington, VA 22217		12. REPORT DATE September 1983
		13. NUMBER OF PAGES 27
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Target Motion Analysis
TMA
Kalman Filter

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This report contains user instructions and program listings for multiple leg Kalman Filter target motion analysis (TMA) procedure for use on the Hewlett-Packard HP-41CV, the Hewlett-Packard HP-75C, the Sharp PC-1500 (Radio Shack TRS-80 PC-2) and the Radio Shack Model 100 portable computers.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

S/N 0102-LF-014-6601

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

A MULTIPLE LEG TMA PROCEDURE WITH PROGRAMS
FOR THE HEWLETT-PACKARD HP-41CV,
THE HEWLETT-PACKARD HP-75C,
THE SHARP PC-1500 (TRS-80 PC-2) AND
THE RADIO SHACK TRS-80 MODEL 100
PORTABLE COMPUTERS.

by

Rex H. Shudde

Naval Postgraduate School
Monterey, California

September 1983

The programs in this report are for use within the Navy, and they are presented without representation or warranty of any kind.

CONTENTS

ABSTRACT

This report contains user instructions and program listings for multiple leg Kalman Filter target motion analysis (TMA) procedure for use on the Hewlett-Packard HP-41CV, the Hewlett-Packard HP-75C, the Sharp PC-1500 (Radio Shack TRS-80 PC-2) and the Radio Shack Model 100 portable computers.

I. INTRODUCTION

This report contains a multiple leg target motion analysis (TMA) program for each of the following portable computers: the Hewlett-Packard HP-41CV, the Hewlett-Packard HP-75C, the Sharp PC-1500 (Radio Shack TRS-80 PC-2) and the Radio Shack TRS-80 Model 100. These programs are part of a set that has been developed for portable computer evaluation.

These programs provide a means of implementing a TMA procedure that is described in Appendix I. The procedure requires at least four bearing observations. It is assumed that during an encounter, the target course and speed remain constant, and that signal transit times are zero.

This algorithm is a Kalman Filter adaption of the least squares TMA algorithm given by Forrest [Ref. 1].

II. THE HEWLETT-PACKARD HP-41CV.

HP-41CV USER INSTRUCTIONS.

There are four user options. These options - called NEW, LEG, FIX and AOP - are used as global labels in the HP-41CV and are assigned to the keys LN, Σ+, 1/x and √x, respectively, for access in the USER mode. Their use is given below. (Memory: SIZE = 40, 706 bytes are required).

	DISPLAY	INSTRUCTION	PRESS
1.		Select USER mode.	USER
2.		**Start a new problem.**	LN (NEW)
3.	BRG ERR?	Key in bearing error.	R/S
4.	OWN CUS?	Key in own course.	R/S
5.	OWN SPD?	Key in own speed.	R/S
6.	TGT BRG?	Key in target bearing.	R/S
7.	TIME?	Key in time (HH.MM).	R/S
8.		Repeat Steps 5 and 6 for each new observation.	
9.		**Input a new leg.**	Σ+ (LEG)
10.	LEG BRG?	Key in new leg bearing.	R/S
11.	LEG DST?	Key in distance on leg.	R/S
12.		Go to Step 4.	
13.		**Compute a fix.**	1/x (FIX)
14.		TGT CUS: (degrees)	R/S
15.		TGT SPD: (knots)	R/S
16.		TGT BRG: (degrees)	R/S
17.		TGT RNG: (meters)	R/S
18.		Next Option?	
19.		**Compute an AOP**	√x (AOP)
20.		SIG 1: (n. mi.)	R/S
21.		SIG 2: (n. mi.)	R/S
22.		ANG: (degrees)	R/S
23.		Next Option?	

HP-41CV SAMPLE PROBLEM

The sensor on your ship has a bearing error of one degree. Your course is 160 deg. at 6 kts. At 1200 hours your sensor detects a target at 350.5 degrees. At 1204 the sensor indicates the target to be at 1.8 deg. To determine a fix, your ship makes a course change during which the course-made-good is 130 deg. and the distance-made-good is 556 meters. Your new course is 080 deg. at 6 kts. The maneuver ends at 1207, at which time the target bearing is 8.3 deg. One more observation places the target at 18.4 deg. at 1211. Estimate the target course, speed, bearing, range, and AOP at 1211 hours.

DISPLAY CONTENTS	USER RESPONSE	COMMENTS
	LN (NEW)	Start a new problem.
BRG ERR?	1 [R/S]	1 deg.
OWN CUS?	160 [R/S]	160 deg.
OWN SPD?	6 [R/S]	6 kts.
TGT BRG?	350.5 [R/S]	350.5 deg.
TIME?	12.00 [R/S]	1200 hours.
TGT BRG?	1.8 [R/S]	1.8 deg.
TIME?	12.04 [R/S]	1204 hours.
TGT BRG?	Σ+ (LEG)	Input a new leg.
LEG BRG?	130 [R/S]	130 deg.
LEG DST?	556 [R/S]	556 meters.
OWN CUS?	080 [R/S]	080 deg.
OWN SPD?	6 [R/S]	6 kts.
TGT BRG?	8.3 [R/S]	8.3 deg.
TIME?	12.07 [R/S]	1207 hours.
TGT BRG?	18.4 [R/S]	18.4 deg.
TIME?	12.11 [R/S]	1211 hours.
TGT BRG?	1/x (FIX)	Compute a fix.
TGT CUS: 123.7	[R/S]	course = 123.7 deg.
TGT SPD: 12.6	[R/S]	speed = 12.6 kts.
TGT BRG: 18.4	[R/S]	bearing = 18.4 deg.
TGT RNG: 4,023.0	[R/S]	range = 4023.0 meters.
NEXT OPTION?	√x (AOP)	Compute an AOP.
SIG 1: 10.40	[R/S]	sigma 1 = 10.40 n. mi.
SIG 2: 0.06	[R/S]	sigma 2 = 0.06 n. mi.
ANG: 18.39	[R/S]	angle = 18.39 deg.
NEXT OPTION?		Quit.

HP-41CV PROGRAM LISTING

```
♦LBL "TMA"
♦LBL 99
TONE 6
"NEXT OPTION?"
AVIEW
GTO 99
♦LBL "BERR"
"BRG ERR?"
PROMPT
D-R
X↑2
STO 23
RTN
GTO 99
♦LBL "NEW"
SF 10
CLRG
1 E3
STO 00
STO 01
STO 02
STO 03
1852
STO 39
XEQ "BERR"
♦LBL "CS"
"OWN CUS?"
PROMPT
STO 26
"OWN SPD?"
PROMPT
STO 27
♦LBL "TB"
"TGT BRG?"
PROMPT
STO 32
1
P-R
STO 25
X<>Y
STO 24
FC? 10
GTO 01
32
RCL 32
X<>Y
P-R
STO 10
X<>Y
STO 11
♦LBL 01
RCL 31
STO 30
"TIME?"
PROMPT
HR
STO 31
FS?C 10
STO 30
RCL 30
-
STO 14
FS?C 09
GTO 00
RCL 26
RCL 27
RCL 14
*
P-R
ST+ 28
X<>Y
ST+ 29
♦LBL 00
RCL 12
RCL 14
*
ST+ 10
RCL 13
RCL 14
*
ST+ 11
RCL 14
ENTER↑
ENTER↑
ENTER↑
RCL 02
*
RCL 05
ENTER↑
+
+
*
```

HP-41CV PROGRAM LISTING (cont.)

```
93 ST+ 00
94 CLX
95 RCL 09
96 *
97 RCL 07
98 +
99 RCL 06
100 +
101 *
102 ST+ 04
103 CLX
104 RCL 03
105 *
106 RCL 08
107 ENTER↑
108 +
109 +
110 *
111 ST+ 01
112 CLX
113 RCL 02
114 *
115 ST+ 05
116 CLX
117 RCL 09
118 *
119 ST+ 06
120 ST+ 07
121 CLX
122 RCL 03
123 *
124 ST+ 08
125 RCL 25
126 RCL 04
127 *
128 RCL 24
129 RCL 00
130 *
131 -
132 STO 19
133 STO 15
134 RCL 25
135 RCL 01
136 *
137 RCL 24
138 RCL 04
139 *
140 -
141 STO 20
142 STO 16
143 RCL 25
144 RCL 07
145 *
146 RCL 24
147 RCL 05
148 *
149 -
150 STO 21
151 STO 17
152 RCL 25
153 RCL 08
154 *
155 RCL 24
156 RCL 06
157 *
158 -
159 STO 22
160 STO 18
161 RCL 25
162 RCL 20
163 *
164 RCL 24
165 RCL 19
166 *
167 -
168 RCL 10
169 RCL 28
170 -
171 X↑2
172 RCL 11
173 RCL 29
174 -
175 X↑2
176 +
177 RCL 23
178 *
179 +
180 ST/ 15
181 ST/ 16
182 ST/ 17
183 ST/ 18
184 RCL 29
```

HP-41CV PROGRAM LISTING (cont.)

```
185 RCL 11
186 -
187 RCL 25
188 *
189 RCL 28
190 RCL 10
191 -
192 RCL 24
193 *
194 -
195 STO 33
196 RCL 15
197 *
198 ST+ 10
199 RCL 33
200 RCL 16
201 *
202 ST+ 11
203 RCL 33
204 RCL 17
205 *
206 ST+ 12
207 RCL 33
208 RCL 18
209 *
210 ST+ 13
211 RCL 15
212 RCL 19
213 *
214 ST- 00
215 RCL 16
216 RCL 20
217 *
218 ST- 01
219 RCL 17
220 RCL 21
221 *
222 ST- 02
223 RCL 18
224 RCL 22
225 *
226 ST- 03
227 RCL 15
228 RCL 20
229 *
230 ST- 04
231 RCL 15
232 RCL 21
233 *
234 ST- 05
235 RCL 15
236 RCL 22
237 *
238 ST- 06
239 RCL 16
240 RCL 21
241 *
242 ST- 07
243 RCL 16
244 RCL 22
245 *
246 ST- 08
247 RCL 17
248 RCL 22
249 *
250 ST- 09
251 GTO "TB"
252♦LBL "LEG"
253 "LEG BRG?"
254 PROMPT
255 "LEG DST?"
256 PROMPT
257 RCL 39
258 /
259 P-R
260 ST+ 28
261 X<>Y
262 ST+ 29
263 SF 09
264 GTO "CS"
265♦LBL "FIX"
266 SF 21
267 FIX 1
268 RCL 13
269 RCL 12
270 R-P
271 STO 34
272 XEQ 91
273 "TGT CUS: "
274 ARCL X
275 AVIEW
276 "TGT SPD: "
```

HP-41CV PROGRAM LISTING (cont.)

```
ARCL 34
AVIEW
RCL 11
RCL 29
-
RCL 10
RCL 28
-
R-P
RCL 39
*
STO 34
XEQ 91
"TGT BRG: "
ARCL X
AVIEW
"TGT RNG: "
ARCL 34
AVIEW
GTO 99
♦LBL "AOP"
SF 21
FIX 2
RCL 00
RCL 01
+
2
/
STO 33
RCL 00
RCL 01
-
2
/
X↑2
RCL 04
X↑2
+
SQRT
STO 34
+
SQRT
"SIG1: "
ARCL X
AVIEW
RCL 33
RCL 34
-
SQRT
"SIG2: "
ARCL X
AVIEW
RCL 04
ST+ X
RCL 00
RCL 01
-
R-P
XEQ 91
2
/
"ANG: "
ARCL X
AVIEW
GTO 99
♦LBL 91
CLX
360
MOD
END
```

III. THE HEWLETT-PACKARD HP-75C.

HP-75C USER INSTRUCTIONS.

There are four user options. These options - labeled 1, 2, 3, and 4 - are used to continue input, input a new leg, compute a fix, and to compute an AOP, respectively. Their use is shown below. (Memory: 2727 bytes for program; 820 bytes for variables. Total = 3547 bytes.)

	DISPLAY	INSTRUCTION	INPUT
1.		**Run Program.**	Run "TMA"
2.	Units:	Select units:	
	1 n.mi.,	Nautical miles,	1
	2 yds., or	yards, or	2
	3 m.?	meters.	3
3.	BRG ERR?	Key in bearing error.	[RTN]
4.	OWN CUS?	Key in own course.	[RTN]
5.	OWN SPD?	Key in own speed.	[RTN]
6.	TGT BRG?	Key in target bearing.	[RTN]
7.	TIME?	Key in time (HH.MM).	[RTN]
8.	1 CONT,	1 returns to Step 6.	1
	2 LEG,	2 goes to Step 9.	2
	3 FIX,	3 goes to Step 13.	3
	4 AOP?	4 goes to Step 18.	4
9.	LEG BRG?	Key in new leg bearing.	[RTN]
10.	LEG DST?	Key in distance on leg.	[RTN]
12.		Go to Step 4.	
13.		TGT CUS: (degrees)	[RTN]
14.		TGT SPD: (knots)	[RTN]
15.		TGT BRG: (degrees)	[RTN]
16.		TGT RNG: (units)	[RTN]
17.		Go to Step 8.	
18.		SIG 1: (units)	[RTN]
19.		SIG 2: (units)	[RTN]
20.		ANG: (degrees)	[RTN]
21.		Go to Step 8.	

HP-75C SAMPLE PROBLEM

The sensor on your ship has a bearing error of one degree. Your course is 160 deg. at 6 kts. At 1200 hours your sensor detects a target at 350.5 degrees. At 1204 the sensor indicates the target to be at 1.8 deg. To determine a fix, your ship makes a course change during which the course-made-good is 130 deg. and the distance-made-good is 556 meters. Your new course is 080 deg. at 6 kts. The maneuver ends at 1207, at which time the target bearing is 8.3 deg. One more observation places the target at 18.4 deg. at 1211. Estimate the target course, speed, bearing, range, and AOP at 1211 hours.

DISPLAY CONTENTS	USER RESPONSE	COMMENTS
Units: 1 n.mi., 2 yds., or 3 m.?	3	Select meters.
BRG ERR?	1 [RTN]	1 deg.
OWN CUS?	160 [RTN]	160 deg.
OWN SPD?	6 [RTN]	6 kts.
TGT BRG?	350.5 [RTN]	350.5 deg.
TIME?	12.00 [RTN]	1200 hours.
1 CONT, 2 LEG, 3 FIX, 4 AOP?	1	Continue input.
TGT BRG?	1.8 [RTN]	1.8 deg.
TIME?	12.04 [RTN]	1204 hours.
1 CONT, 2 LEG, 3 FIX, 4 AOP?	2	Input new leg.
LEG BRG?	130 [RTN]	130 deg.
LEG DST?	556 [RTN]	556 meters.
OWN CUS?	080 [RTN]	080 deg.
OWN SPD?	6 [RTN]	6 kts.
TGT BRG?	8.3 [RTN]	8.3 deg.
TIME?	12.07 [RTN]	1207 hours.
1 CONT, 2 LEG, 3 FIX, 4 AOP?	1	Continue input.
TGT BRG?	18.4 [RTN]	18.4 deg.
TIME?	12.11 [RTN]	1211 hours.
1 CONT, 2 LEG, 3 FIX, 4 AOP?	3	Compute a fix.
CUS = 123.7		course = 123.7 deg.
SPD = 12.6		speed = 12.6 kts.
BRG = 18.4		bearing = 18.4 deg.
RNG = 4023 m.		range = 4023 meters.
1 CONT, 2 LEG, 3 FIX, 4 AOP?	4	Compute an AOP.
σ 1 = 19263.71 m.		sigma 1 = 19263.71 m.
σ 2 = 105.22 m.		sigma 2 = 105.22 m.
ANG = 18.4		angle = 18.4 deg.
1 CONT, 2 LEG, 3 FIX, 4 AOP?		Quit.

HP-75C PROGRAM LISTING

```
10 REM TMA
20 GOTO 60
30 C$=KEY$ @ IF C$='' TH
EN 30
40 FOR C=1 TO LEN(C0$) @
 IF C$=C0$[C,C] THEN RET
URN
50 NEXT C @ GOTO 30
60 OPTION BASE 1 @ OPTIO
N ANGLE DEGREES
70 DIM P(10),H(4),X(4)
80 DEF FNR(X,M) = INT(M*
X+.5)/M
90 X0=0 @ Y0=0 @ F5=0
100 F0=1 @ T9=0 @ L=0 @
FOR I=1 TO 4 @ FOR J=I T
O 4 @ L=L+1 @ T=0 @ IF I
=J THEN T=1000
110 P(L)=T @ NEXT J @ NE
XT I
120 DISP 'Units: 1 n.mi,
 2 yds, or 3 m.? ' @ C0$
='123' @ GOSUB 30 @ U=C
130 IF U=1 THEN K=1 @ U$
='nm' @ GOTO 160
140 IF U=2 THEN K=2025 @
 U$='yds.' @ GOTO 160
150 IF U=3 THEN K=1852 @
 U$='m.' @ GOTO 160
160 GOSUB 4000 @ GOSUB 4
010 @ GOSUB 4020 @ GOSUB
 4030 @ GOSUB 4040
170 GOTO 240
180 DISP '1 CONT, 2 LEG,
 3 FIX, 4 AOP?' @ C0$='1
234' @ GOSUB 30
190 IF C=2 THEN GOSUB 40
70 @ GOTO 220
200 IF C=3 THEN GOSUB 10
00 @ GOTO 180
210 IF C=4 THEN GOSUB 10
50 @ GOTO 180
220 IF F5=1 THEN GOSUB 4
010
230 GOSUB 4020 @ GOSUB 4
040
240 X(1)=X(1)+X(3)*D9 @
X(2)=X(2)+X(4)*D9
250 P(1)=P(1)+(2*P(3)+P(
8)*D9)*D9
260 P(2)=P(2)+(P(4)+P(6)
+P(9)*D9)*D9
270 P(5)=P(5)+(2*P(7)+P(
10)*D9)*D9
280 P(3)=P(3)+P(8)*D9
290 P(4)=P(4)+P(9)*D9
300 P(6)=P(6)+P(9)*D9
310 P(7)=P(7)+P(10)*D9
320 H(1)=C*P(2)-S*P(1)
330 H(2)=C*P(5)-S*P(2)
340 H(3)=C*P(6)-S*P(3)
350 H(4)=C*P(7)-S*P(4)
360 D=C*H(2)-S*H(1)+E*((
X(1)-X0)^2+(X(2)-Y0)^2)
370 Z=Y0*C-X0*S @ S5=Z-(
C*X(2)-S*X(1))
380 FOR I=1 TO 4 @ X(I)=
X(I)+H(I)/D*S5 @ NEXT I
390 L=0 @ FOR I=1 TO 4 @
 FOR J=I TO 4 @ L=L+1 @
P(L)=P(L)-H(I)*H(J)/D @
NEXT J @ NEXT I
400 GOTO 180
1000 X2=X(1)-X0 @ Y2=X(2
)-Y0 @ R2=SQR(X2*X2+Y2*Y
2) @ B2=ANGLE(X2,Y2) @ I
F B2<0 THEN B2=B2+360
1010 X2=X(3) @ Y2=X(4) @
 S2=SQR(X2*X2+Y2*Y2) @ C
2=ANGLE(X2,Y2) @ IF C2<0
 THEN C2=C2+360
1020 DISP 'CUS =';FNR(C2
,10); @ DISP ' SPD =';FN
R(S2,10) @ GOSUB 4100
1030 DISP 'BRG =';FNR(B2
,10); @ DISP ' RNG =';FN
R(R2*K,10);U$ @ GOSUB 41
00
1040 RETURN
1050 A=P(1) @ B=P(5) @ D
=P(2) @ A1=ANGLE(A-B,2*D
)/2 @ IF A1<0 THEN A1=A1
+180
1060 S3=(A+B)/2 @ C3=SQR
((A-B)^2/4+D*D) @ S1=S3+
C3
```

HP-75C PROGRAM LISTING (cont.)

```
1070 S2=S3-C3
1080 DISP ' 1 = ';FNR(K
*SQR(S1),100);U$ @ GOSUB
 4100
1090 DISP '   2 = ';FNR
(K*SQR(S2),100);U$ @ GOS
UB 4100
1100 DISP 'ANG = ';FNR(A
1,10) @ GOSUB 4100 @ RET
URN
4000 INPUT 'BRG ERR? ';
E @ E=RAD(E) @ E=E*E @ R
ETURN
4010 INPUT 'OWN CUS? ';
C0 @ INPUT 'OWN SPD? ';
S0 @ RETURN
4020 INPUT 'TGT BRG? ';
B9 @ S=SIN(B9) @ C=COS(B
9) @ RETURN
4030 R9=32 @ X(1)=R9*C @
 X(2)=R9*S @ X(3)=0 @ X(
4)=0 @ RETURN @ REM R9 =
  EST RNG
4040 T8=T9 @ INPUT 'TIME
? '; T9 @ T9=FNH(T9) @ I
F F0=1 THEN F0=0 @ T8=T9
4050 D9=T9-T8 @ IF F5=0
THEN X0=X0+D9*S0*COS(C0)
 @ Y0=Y0+D9*S0*SIN(C0)
4060 F5=0 @ RETURN
4070 INPUT 'LEG BRG? ';
B8 @ B8=FNH(B8) @ INPUT
'LEG DST? '; R8 @ R8=R8/
K
4080 X0=X0+R8*COS(B8) @
Y0=Y0+R8*SIN(B8)
4090 F5=1 @ RETURN
4100 WAIT 5 @ RETURN
4110 DEF FNH(V)
4120 H=INT(V) @ V=FP(V)*
100 @ FNH=(100*FP(V)/60+
INT(V))/60+H
4130 END DEF
```

IV. THE SHARP PC-1500 (RADIO SHACK TRS-80 PC-2).

SHARP PC-1500 (TRS-80 PC-2) USER INSTRUCTIONS.

There are three main user options. These options - labeled 1, 2, and 3 - are used to continue input, input a new leg, and to compute a fix, respectively. Other options include unit selection (n. mi., yds. or meters) and AOP selection. Their use is shown below. (Memory: 2050 bytes for program; number of bytes for variables is not available.)

	DISPLAY	INSTRUCTION	INPUT
1.		**Run Program.**	RUN [ENTER]
2.	Units:	Select units:	
	1 n.mi.,	Nautical miles,	1
	2 yds., or	yards, or	2
	3 meter?	meters.	3
3.	BRG ERR?	Key in bearing error.	[ENTER]
4.	OWN CUS?	Key in own course.	[ENTER]
5.	OWN SPD?	Key in own speed.	[ENTER]
6.	TGT BRG?	Key in target bearing.	[ENTER]
7.	TIME?	Key in time (HH.MM).	[ENTER]
8.	1 CONT,	1 returns to Step 6.	1
	2 LEG,	2 goes to Step 9.	2
	3 FIX?	3 goes to Step 13.	3
9.	LEG BRG?	Key in new leg bearing.	[ENTER]
10.	LEG DST?	Key in distance on leg.	[ENTER]
12.		Go to Step 4.	
13.		TGT CUS: (degrees)	[ENTER]
14.		TGT SPD: (knots)	[ENTER]
15.		TGT BRG: (degrees)	[ENTER]
16.		TGT RNG: (units)	[ENTER]
17.	AOP? Y or N		
		Y goes to Step 18.	Y
		N goes to Step 8.	N
18.		SIG 1: (units)	[ENTER]
19.		SIG 2: (units)	[ENTER]
20.		ANG: (degrees)	[ENTER]
21.		Go to Step 8.	

SHARP PC-1500 (TRS-80 PC-2) SAMPLE PROBLEM

The sensor on your ship has a bearing error of one degree. Your course is 160 deg. at 6 kts. At 1200 hours your sensor detects a target at 350.5 degrees. At 1204 the sensor indicates the target to be at 1.8 deg. To determine a fix, your ship makes a course change during which the course-made-good is 130 deg. and the distance-made-good is 556 meters. Your new course is 080 deg. at 6 kts. The maneuver ends at 1207, at which time the target bearing is 8.3 deg. One more observation places the target at 18.4 deg. at 1211. Estimate the target course, speed, bearing, range, and AOP at 1211 hours.

DISPLAY CONTENTS	USER RESPONSE	COMMENTS
Units: 1 n.mi., 2 yds., or 3 m.?	3	Select meters.
BRG ERR?	1 [ENTER]	1 deg.
OWN CUS?	160 [ENTER]	160 deg.
OWN SPD?	6 [ENTER]	6 kts.
TGT BRG?	350.5 [ENTER]	350.5 deg.
TIME?	12.00 [ENTER]	1200 hours.
1 CONT, 2 LEG, 3 FIX?	1	Continue input.
TGT BRG?	1.8 [ENTER]	1.8 deg.
TIME?	12.04 [ENTER]	1204 hours.
1 CONT, 2 LEG, 3 FIX?	2	Input new leg.
LEG BRG?	130 [ENTER]	130 deg.
LEG DST meters?	556 [ENTER]	556 meters.
OWN CUS?	080 [ENTER]	080 deg.
OWN SPD?	6 [ENTER]	6 kts.
TGT BRG?	8.3 [ENTER]	8.3 deg.
TIME?	12.07 [ENTER]	1207 hours.
1 CONT, 2 LEG, 3 FIX?	1	Continue input.
TGT BRG?	18.4 [ENTER]	18.4 deg.
TIME?	12.11 [ENTER]	1211 hours.
1 CONT, 2 LEG, 3 FIX?	3	Compute a fix.
CUS = 123.7		course = 123.7 deg.
SPD = 12.6		speed = 12.6 kts.
BRG = 18.4		bearing = 18.4 deg.
RNG = 4023 m.		range = 4023 meters.
AOP? Y or N	Y	Compute an AOP.
SIG1 = 19263.71 m.		sigma 1 = 19263.71 m.
SIG2 = 105.22 m.		sigma 2 = 105.22 m.
ANG = 18.39		angle = 18.39 deg.
1 CONT, 2 LEG, 3 FIX?		Quit.

SHARP PC-1500 (TRS-80 PC-2) PROGRAM LISTING

```
  10:REM  TMA
  15:WAIT :GOTO 30
  20:C$=INKEY$ :IF
     C$=""THEN 20
  22:FOR C=1TO LEN
     (C0$):IF C$=
     MID$ (C0$,C,1)
     THEN RETURN
  24:NEXT C:GOTO 20
  30:DIM P(10),H(4)
     ,X(4):RD=PI /1
     80:DEGREE
  50:X0=0:Y0=0:F5=0
  60:F0=1:T9=0:L=0:
     FOR I=1TO 4:
     FOR J=1TO 4:L=
     L+1:T=0:IF I=J
     THEN LET T=100
     0
  70:P(L)=T:NEXT J:
     NEXT I
  80:PAUSE "1 n.mi,
      2 yds, or 3 m
     eter?"
  85:C0$="123":
     GOSUB 20:K=C
  90:IF K=1THEN LET
     U$=" n.mi.":
     GOTO 130
 100:IF K=2THEN LET
     K=2025:U$=" yd
     s.":GOTO 130
 110:IF K=3THEN LET
     K=1852:U$=" me
     ters"
 130:GOSUB 4000:
     GOSUB 4010:
     GOSUB 4020:
     GOSUB 4030:
     GOSUB 4040
 140:GOTO 210
 150:PAUSE "1 CONT,
      2 LEG, 3 FIX?
      ":C0$="123":
     GOSUB 20:CLS :
     WAIT
 160:IF C=2THEN
     GOSUB 4070:
     GOTO 190
 170:IF C=3THEN
     GOSUB 1000:
     GOTO 150
 190:IF F5=1THEN
     GOSUB 4010
 200:GOSUB 4020:
     GOSUB 4040
 210:X(1)=X(1)+X(3)
     *D9:X(2)=X(2)+
     X(4)*D9
 220:P(1)=P(1)+(2*P
     (3)+P(8)*D9)*D
     9
 230:P(2)=P(2)+(P(4
     )+P(6)+P(9)*D9
     )*D9
 240:P(5)=P(5)+(2*P
     (7)+P(10)*D9)*
     D9
 250:P(3)=P(3)+P(8)
     *D9
 260:P(4)=P(4)+P(9)
     *D9
 270:P(6)=P(6)+P(9)
     *D9
 280:P(7)=P(7)+P(10
     )*D9
 290:H(1)=C*P(2)-S*
     P(1)
 300:H(2)=C*P(5)-S*
     P(2)
 310:H(3)=C*P(6)-S*
     P(3)
 320:H(4)=C*P(7)-S*
     P(4)
 330:D=C*H(2)-S*H(1
     )+E*((X(1)-X0)
     ^2+(X(2)-Y0)^2
     )
 340:Z=Y0*C-X0*S:S5
     =Z-(C*X(2)-S*X
     (1))
 350:FOR I=1TO 4:X(
     I)=X(I)+H(I)/D
     *S5:NEXT I
 360:L=0:FOR I=1TO
     4:FOR J=1TO 4:
     L=L+1:P(L)=P(L
     )-H(I)*H(J)/D:
     NEXT J:NEXT I
 370:GOTO 150
1000:X=X(1)-X0:Y=
     X(2)-Y0:R2=
     SQR (X*X+Y*Y
     ):GOSUB 5000
     :B2=AT
```

```
1010:X=X(3):Y=X(4
     ):S2=SQR (X*
     X+Y*Y):GOSUB
     5000:C2=AT
1020:M=10:X=C2:
     GOSUB 5010:
     PRINT "TGT C
     US =";X:X=S2
     :GOSUB 5010:
     PRINT "TGT S
     PD =";X
1030:X=B2:GOSUB 5
     010:PRINT "T
     GT BRG =";X:
     X=R2*K:GOSUB
     5010:PRINT "
     TGT RNG =";X
     ;U$
1040:PAUSE "AOP?
     Y or N ":C0$
     ="YyNn":
     GOSUB 20:CLS
1050:IF C>2THEN
     RETURN
1080:A=P(1):B=P(5
     ).D=P(2):X=A
     -B.Y=2*D:
     GOSUB 5000:A
     1=AT/2
1090:S3=(A+B)/2:C
     3=SQR ((A-B)
     ^2/4+D*D)
1100:S1=S3+C3:S2=
     S3-C3
1110:M=100.X=K*
     SQR (S1):
     GOSUB 5010:
     PRINT "SIG1=
     ";X;U$
1120:X=K*SQR (S2)
     :GOSUB 5010:
     PRINT "SIG2
     =";X;U$
1130:X=A1:GOSUB 5
     010:PRINT "A
     NG =";X:
     RETURN
```

```
4000:INPUT "BRG E
     RR? ";E:E=E*
     RD:E=E*E:
     RETURN
4010:INPUT "OWN C
     US? ";C0:
     INPUT "OWN S
     PD? ";S0:
     RETURN
4020:INPUT "TGT B
     RG? ";B9:S=
     SIN B9:C=COS
     B9:RETURN
4030:R9=32:X(1)=R
     9*C:X(2)=R9*
     S:X(3)=0:X(4
     )=0:RETURN
4040:T8=T9:INPUT
     "TIME? ";T9:
     CLS :T9=DEG
     T9:IF F0=1
     THEN LET F0=
     0:T8=T9
4050:D9=T9-T8:IF
     F5=0THEN LET
     X0=X0+D9*S0*
     COS C0:Y0=Y0
     +D9*S0*SIN C
     0
4060:F5=0:RETURN
4070:INPUT "LEG B
     RG? ";B8:B8=
     DEG B8
4075:WAIT 0:PRINT
     "LEG DST ";U
     $;:INPUT "?
     ";R8:PAUSE :
     R8=R8/K
4080:X0=X0+R8*COS
     B8:Y0=Y0+R8*
     SIN B8
4090:F5=1:RETURN
5000:AT=ATN (Y/(X
     +1E-9*(0=X))
     )+180*(X<0):
     AT=AT+360*(A
     T<0):RETURN
5010:X=INT (M*X+.
     5)/M:RETURN
```

V. THE RADIO SHACK TRS-80 MODEL 100.

RADIO SHACK TRS-80 MODEL 100 USER INSTRUCTIONS.

There are four user options. These options - labeled 1, 2, 3, and 4 - are used to continue input, input a new leg, compute a fix, and to compute an AOP, respectively. Their use is shown below. (Memory: 2503 bytes for program; 815 bytes for variables. Total = 3318 bytes.)

	DISPLAY	INSTRUCTION	INPUT
1.		**Run Program.**	Run [ENTER]
2.	Units:	Select units:	
	1 n.mi.,	Nautical miles,	1
	2 yds., or	yards, or	2
	3 m.?	meters?	3
3.	BRG ERR?	Key in bearing error.	[ENTER]
4.	OWN CUS?	Key in own course.	[ENTER]
5.	OWN SPD?	Key in own speed.	[ENTER]
6.	TGT BRG?	Key in target bearing.	[ENTER]
7.	TIME?	Key in time (HH.MM).	[ENTER]
8.	1 Continue,	1 returns to Step 6.	1
	2 Leg,	2 goes to Step 9.	2
	3 Fix, or	3 goes to Step 13.	3
	4 AOP?	4 goes to Step 18.	4
9.	LEG BRG?	Key in new leg bearing.	[ENTER]
10.	LEG DST?	Key in distance on leg.	[ENTER]
12.		Go to Step 4.	
13.		TGT CUS: (degrees)	[ENTER]
14.		TGT SPD: (knots)	[ENTER]
15.		TGT BRG: (degrees)	[ENTER]
16.		TGT RNG: (units)	[ENTER]
17.		Press any key to continue.	[ENTER]
		Go to Step 8.	
18.		SIG 1: (units)	[ENTER]
19.		SIG 2: (units)	[ENTER]
20.		ANG: (degrees)	[ENTER]
21.		Press any key to continue.	[ENTER]
		Go to Step 8.	

RADIO SHACK TRS-80 MODEL 100 SAMPLE PROBLEM

The sensor on your ship has a bearing error of one degree. Your course is 160 deg. at 6 kts. At 1200 hours your sensor detects a target at 350.5 degrees. At 1204 the sensor indicates the target to be at 1.8 deg. To determine a fix, your ship makes a course change during which the course-made-good is 130 deg. and the distance-made-good is 556 meters. Your new course is 080 deg. at 6 kts. The maneuver ends at 1207, at which time the target bearing is 8.3 deg. One more observation places the target at 18.4 deg. at 1211. Estimate the target course, speed, bearing, range, and AOP at 1211 hours.

DISPLAY CONTENTS	USER RESPONSE	COMMENTS
Units: 1 n.mi., 2 yds., or 3 m.?	3	Select meters.
BRG ERR?	1 [ENTER]	1 deg.
OWN CUS?	160 [ENTER]	160 deg.
OWN SPD?	6 [ENTER]	6 kts.
TGT BRG?	350.5 [ENTER]	350.5 deg.
TIME?	12.00 [ENTER]	1200 hours.
1 Continue, 2 Leg, 3 Fix, or 4 AOP?	1	Continue input.
TGT BRG?	1.8 [ENTER]	1.8 deg.
TIME?	12.04 [ENTER]	1204 hours.
1 Continue, 2 Leg, 3 Fix, or 4 AOP?	2	Input new leg.
LEG BRG?	130 [ENTER]	130 deg.
LEG DST meters?	556 [ENTER]	556 meters.
OWN CUS?	080 [ENTER]	080 deg.
OWN SPD?	6 [ENTER]	6 kts.
TGT BRG?	8.3 [ENTER]	8.3 deg.
TIME?	12.07 [ENTER]	1207 hours.
1 Continue, 2 Leg, 3 Fix, or 4 AOP?	1	Continue input.
TGT BRG?	18.4 [ENTER]	18.4 deg.
TIME?	12.11 [ENTER]	1211 hours.
1 Continue, 2 Leg, 3 Fix, or 4 AOP?	3	Compute a fix.
CUS = 123.7		course = 123.7 deg.
SPD = 12.6		speed = 12.6 kts.
BRG = 18.4		bearing = 18.4 deg.
RNG = 4023 m.		range = 4023 meters.
1 Continue, 2 LEG, 3 Fix, or 4 AOP?	4	Compute an AOP.
SIG 1 = 19263.71 m.		sigma 1 = 19263.71 m.
SIG 2 = 105.22 m.		sigma 2 = 105.22 m.
ANG = 18.39		angle = 18.39 deg.
1 Continue, 2 LEG, 3 Fix, or 4 AOP?		Quit.

TRS-80 MODEL 100 PROGRAM LISTING

```
10 REM TMA
20 DIMP(4,4),H(4),X(4):PI=4*ATN(1):TP=PI+PI:RD=PI/180:GOTO50
25 FORC=1TO9:C$=INKEY$:NEXTC
26 PRINT:PRINT"Press any key to continue."
27 C$=INKEY$:IFC$=""GOTO27
28 RETURN
32 FORC=1TO9:C$=INKEY$:NEXTC
33 C$=INKEY$:IFC$=""GOTO33
34 C=ASC(C$):C$=CHR$(C+32*(C>90))
35 FORC=1TOLEN(C0$):IFC$=MID$(C0$,C,1)THENRETURN
36 NEXTC:GOTO33
40 X=INT(M*X+.5)/M:RETURN:REM ROUND
50 X0=0:Y0=0:F5=0
60 F0=1:T9=0:FORI=1TO4:FORJ=1TO4:T=0:IFI=JTHENT=1000
70 P(I,J)=T:NEXTJ:NEXTI
80 CLS:PRINT"Units:":PRINT"  1 - n.mi.,"
82 PRINT"  2 - yds, or":PRINT"  3 - meters?"
85 C0$="123":GOSUB32:K=C
90 IFK=1THENU$=" n.mi.":GOTO130
100 IFK=2THENK=2025:U$=" yds.":GOTO130
110 IFK=3THENK=1852:U$=" meters."
130 GOSUB4000:GOSUB4010:GOSUB4020:GOSUB4030:GOSUB4040
140 GOTO210
150 CLS:PRINT"Select Option:":PRINT"  1 - Continue,"
152 PRINT"  2 - New Leg,":PRINT"  3 - Fix, or":PRINT"  4 - AOP?"
153 C0$="1234":GOSUB32:CLS
160 IFC=2THENGOSUB4070:GOTO190
170 IFC=3THENGOSUB1000:GOTO150
180 IFC=4THENGOSUB1080:GOTO150
190 IFF5=1THENGOSUB4010
200 GOSUB4020:GOSUB4040
210 X(1)=X(1)+X(3)*D9:X(2)=X(2)+X(4)*D9
220 P(1,1)=P(1,1)+(2*P(1,3)+P(3,3)*D9)*D9
230 P(1,2)=P(1,2)+(P(1,4)+P(2,3)+P(3,4)*D9)*D9
240 P(2,2)=P(2,2)+(2*P(2,4)+P(4,4)*D9)*D9
250 P(1,3)=P(1,3)+P(3,3)*D9
260 P(1,4)=P(1,4)+P(3,4)*D9
270 P(2,3)=P(2,3)+P(3,4)*D9
280 P(2,4)=P(2,4)+P(4,4)*D9
290 FORI=1TO4:FORJ=ITO4:P(J,I)=P(I,J):NEXTJ:NEXTI
```

```
300 H(1)=C*P(1,2)-S*P(1,1)
310 H(2)=C*P(2,2)-S*P(2,1)
320 H(3)=C*P(3,2)-S*P(3,1)
330 H(4)=C*P(4,2)-S*P(4,1)
340 D=C*H(2)-S*H(1)+E*((X(1)-X0)^2+(X(2)-Y0)^2)
350 Z=Y0*C-X0*S:S5=Z-(C*X(2)-S*X(1))
360 FORI=1TO4:X(I)=X(I)+H(I)/D*S5:NEXTI
370 FORI=1TO4:FORJ=1TO4:P(I,J)=P(I,J)-H(I)*H(J)/D:NEXTJ:NEXTI
380 GOTO150
1000 X=X(1)-X0:Y=X(2)-Y0:R2=SQR(X*X+Y*Y):GOSUB5000:B2=AT
1010 X=X(3):Y=X(4):S2=SQR(X*X+Y*Y):GOSUB5000:C2=AT
1020 M=10:X=C2/RD:GOSUB40:PRINT"TGT CUS
=";X:X=S2:GOSUB40:PRINT"TGT SPD =";X
1030 X=B2/RD:GOSUB40:PRINT"TGT BRG =";X:X=R2*K:GOSUB40:PRINT"TGT
RNG =";X;U$
1040 GOSUB25:RETURN
1080 A=P(1,1):B=P(2,2):D=P(1,2):X=A-B:Y=D+D:GOSUB5000:A1=AT/2
1090 S3=(A+B)/2:C3=SQR((A-B)^2/4+D*D):S1=S3+C3
1100 S2=S3-C3
1110 M=100:X=K*SQR(S1):GOSUB40:PRINT"SIG 1 =";X;U$
1120 X=K*SQR(S2):GOSUB40:PRINT"SIG 2 =";X;U$
1130 X=A1/RD:GOSUB40:PRINT"ANG =";X:GOSUB25:RETURN
4000 CLS:INPUT"BRG ERR";E:E=E*RD:E=E*E:RETURN
4010 INPUT"OWN CUS";C0:C0=C0*RD:INPUT"OWN SPD";S0:RETURN
4020 INPUT"TGT BRG";B9:B9=B9*RD:S=SIN(B9):C=COS(B9):RETURN
4030 R9=32:X(1)=R9*C:X(2)=R9*S:X(3)=0:X(4)=0:RETURN
4040
T8=T9:INPUT"TIME";T9:CLS:X=T9:GOSUB4110:T9=V:IFF0=1THENF0=0:T8=T9

4050 D9=T9-T8:IFF5=0THENX0=X0+D9*S0*COS(C0):Y0=Y0+D9*S0*SIN(C0)
4060 F5=0:RETURN
4070 INPUT"LEG BRG";B8:B8=B8*RD:PRINT"LEG DST";U$;:INPUTR8:R8=R8/
K
4080 X0=X0+R8*COS(B8):Y0=Y0+R8*SIN(B8)
4090 F5=1:RETURN
4110 SS=SGN(X):X=ABS(X):H=INT(X):M0=1:GOSUB4200:V=X*
100:X=V:GOSUB4200:
4120 V=SS*((100*X/60+INT(V))/60+H):RETURN
4200 X=X-M0*INT(X/M0):RETURN:REM MOD FCTN
5000 AT=ATN(Y/(X-1E-9*(X=0)))-PI*(X<0):AT=AT-TP*(AT<0):RETURN
```

VI. REFERENCES

1. R. N. Forrest, "Programs for a Multiple Leg Target Motion Analysis Procedure", Technical Report NPS55-82-026, October 1982, Naval Postgraduate School, Monterey, CA 93940.

2. A. Gelb, editor, Applied Optimal Estimation, The M.I.T. Press, 1974.

Appendix I: The Fixing Algorithm

The Kalman Filter TMA model may be developed as follows: A target and an observer move on a plane surface. The observer, at position (u_k, w_k), measures the target bearing B_k at time t_k, for $k = 1, \ldots, n$. The rectangular coordinate system used to estimate the target's position is shown in Figure 1.



Figure 1. The encounter geometry. The north-south coordinates x and u and the east-west coordinates y and w are measured from the same origin.

In Figure 1, the point labeled (u_k, w_k) represents the tracker's position and the point labeled (x_k, y_k) represents the true target position at t_k, the time associated with an observed bearing B_k. At time t_k, B_{Tk} represents the true target bearing and r_k represents the true target range. The bearing error associated with an observation is assumed to be distributed $N(\emptyset, \sigma_k^2)$. The error e_k, at a range r_k, is then $N(\emptyset, r_k^2\sigma_k^2)$.

From Figure 1,

$$e_k = r_k \sin (B_{Tk} - B_k),$$

which rearranges to

$$e_k = u_k \sin B_k - w_k \cos B_k - x_k \sin B_k + y_k \cos B_k. \tag{1}$$

In the least squares procedure [Ref. 1], one sets

$$\begin{aligned} x_k &= x_\emptyset + v_x(t_k - t_\emptyset), \\ y_k &= y_\emptyset + v_y(t_k - t_\emptyset), \end{aligned} \tag{2}$$

and then minimizes Σe_k^2 with respect to $x_\emptyset$, $y_\emptyset$, v_x and v_y, where $(x_\emptyset, y_\emptyset)$ represents the target's position and (v_x, v_y) represents the target's velocity at time $t_\emptyset$. The minimizing values are the estimated position and velocity, denoted by $(\hat{x}_\emptyset, \hat{y}_\emptyset)$ and $(\hat{v}_x, \hat{v}_y)$, respectively.

An identical least squares model is obtained if one sets

$$z_k = x_k \sin B_k - y_k \cos B_k + e_k, \tag{3}$$

where z_k is computed from

$$z_k = u_k \sin B_k - w_k \cos B_k. \qquad (4)$$

Instead of substituting Equ.(2) into Equ.(3) and using the least squares procedure, we treat Equ.(3) as the measurement model in a Kalman Filter [Ref. 2]. The system model is taken to be

$$\mathbf{x}_k = \Phi_{k-1}\mathbf{x}_{k-1}, \quad \text{for } k = 1, 2, \ldots.$$

where

$$\Phi_k = \begin{bmatrix} 1 & 0 & T_k & 0 \\ 0 & 1 & 0 & T_k \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

and

$$\mathbf{x}_k = [x_k \ \ y_k \ \ v_x \ \ v_y]'.$$

In these equations, $T_k = t_{k+1} - t_k$ and [...]' denotes the transpose matrix.

The measurement model (Equ. 3) can be written as

$$z_k = H_k\mathbf{x}_k + e_k,$$

where $e_k = N(0, R_k)$, $R_k = r_k^2\sigma_k^2$, and

$$H_k = [\sin B_k \ \ -\cos B_k \ \ 0 \ \ 0].$$

If we let (˜) denote an extrapolated quantity (the movement phase of the Kalman Filter) and (^) denote and updated estimate (following an observation), then, for the extrapolation

$$\tilde{x}_k = \Phi_{k-1}\hat{x}_{k-1}, \text{ and}$$
$$\tilde{P}_k = \Phi_{k-1}\hat{P}_{k-1}\Phi'_{k-1}.$$

For the measurement update,

$$K_k = \tilde{P}_k H'_k [H_k \tilde{P}_k H'_k + R_k]^{-1},$$

$$\hat{x}_k = \tilde{x}_k + K_k[z_k - H_k\tilde{x}_k], \text{ and}$$

$$\hat{P}_k = \tilde{P}_k - K_k(\tilde{P}_k H'_k)'.$$

Since the true range r_k is not known, we approximate R_k by $R_k = \hat{r}_k^2\sigma_k^2$ where

$$\hat{r}_k^2 = (\hat{x}_k - u_k)^2 + (\hat{y}_k - w_k)^2.$$

To initialize the filter, we assume the initial range of the target to be one convergence zone (CZ), about 32 n. mi., with a standard deviation of one CZ, and the initial position is in the direction of the first bearing observation. Further, the initial speed is taken to be zero with a standard deviation of 32 knots. Thus, we set

$$\tilde{x}_1 = [32 \sin B_1 \quad -32 \cos B_1 \quad 0 \quad 0]',$$

and the covariance matrix

$$\tilde{P}_1 = \begin{bmatrix} 1000 & 0 & 0 & 0 \\ 0 & 1000 & 0 & 0 \\ 0 & 0 & 1000 & 0 \\ 0 & 0 & 0 & 1000 \end{bmatrix}$$

Following the first observation we find $\hat{\mathbf{x}}_1 = \mathbf{x}_1$.

In this model the observer's position (u_k, w_k) at time t_k is taken to be

$$u_k = u_0 + v_x T_k \cos C_s,$$

and,

$$w_k = w_0 + v_y T_k \sin C_s,$$

where (u_0, w_0) is the observer's position at time t_0. It would be easy to modify this algorithm to allow for multiple sensors by generating the sensor's coordinates (u_k, w_k) at time t_k.

If the elements of the covariance matrix $\hat{P}_k$ are denoted by $(\hat{p}_{ij})$ then the semimajor and semiminor axes, σ_1 and σ_2, of the AOP are computed as follows:

$$\sigma_1^2 = (\hat{p}_{11} + \hat{p}_{22})/2 + [(\hat{p}_{11} - \hat{p}_{22})^2/4 + \hat{p}_{12}^2]^{1/2},$$

$$\sigma_2^2 = (\hat{p}_{11} + \hat{p}_{22})/2 - [(\hat{p}_{11} - \hat{p}_{22})^2/4 + \hat{p}_{12}^2]^{1/2},$$

where the orientation angle A of the semimajor axis of the AOP is given by

$$A = 0.5 \text{ qatn } (2\hat{p}_{12}, \hat{p}_{11} - \hat{p}_{22}).$$

The notation qatan(Y,X) denotes arctan(Y/X) corrected for quadrant.

Appendix II: Some Computational Details

Let the elements of the covariance matrices $\hat{P}$ and $\tilde{P}$ be denoted by $(\hat{p}_{ij})$ and $(\tilde{p}_{ij})$, respectively. Then, suppressing the time dependent subscripts, the covariance extrapolation,

$$\tilde{P} = \Phi\hat{P}\Phi'$$

can be expanded into the following components:

$$\begin{aligned}
\tilde{p}_{11} &= \hat{p}_{11} + 2\hat{p}_{13}T + \hat{p}_{33}T^2 \\
\tilde{p}_{22} &= \hat{p}_{22} + 2\hat{p}_{24}T + \hat{p}_{44}T^2 \\
\tilde{p}_{12} &= \hat{p}_{12} + (\hat{p}_{14} + \hat{p}_{23})T + \hat{p}_{34}T^2 \\
\tilde{p}_{13} &= \hat{p}_{13} + \hat{p}_{33}T \\
\tilde{p}_{14} &= \hat{p}_{14} + \hat{p}_{34}T \\
\tilde{p}_{23} &= \hat{p}_{23} + \hat{p}_{34}T \\
\tilde{p}_{24} &= \hat{p}_{24} + \hat{p}_{44}T \\
\tilde{p}_{33} &= \hat{p}_{33} \\
\tilde{p}_{34} &= \hat{p}_{34} \\
\tilde{p}_{34} &= \hat{p}_{34}
\end{aligned}$$

Let the elements of $\tilde{P}H'$ be denoted by (h_i). Then,

$$\begin{aligned}
h_1 &= \tilde{p}_{12} \cos B - \tilde{p}_{11} \sin B \\
h_2 &= \tilde{p}_{22} \cos B - \tilde{p}_{12} \sin B \\
h_3 &= \tilde{p}_{23} \cos B - \tilde{p}_{13} \sin B \\
h_4 &= \tilde{p}_{24} \cos B - \tilde{p}_{14} \sin B
\end{aligned}$$

and,

$$H\tilde{P}H' = h_2 \cos B \; - \; h_1 \sin B.$$

Let the elements of K be denoted by (k_i). Then,

$$k_i = h_i/D \quad \text{for } i = 1, 2, 3, 4$$

where $D = H\tilde{P}H' + R$.

Finally,

$$\hat{p}_{ij} = \tilde{p}_{ij} - k_i h_j \quad \text{for } i, j = 1, 2, 3, 4.$$

DISTRIBUTION LIST

	NO. OF COPIES
Defense Technical Information Center Cameron Station Alexandria, VA 22314	2
Library, Code 0142 Naval Postgraduate School Monterey, CA 93940	2
Office of Research Administration Code 012A Naval Postgraduate School Monterey, CA 93940	1
Library, Code 55 Naval Postgraduate School Monterey, CA 93940	1
Office of Naval Research Fleet Activity Support Division Code ONR-230 800 North Quincy Street Arlington, VA 22217	2
Navy Tactical Support Activiity P.O. Box 1042 Silver Springs, MD 20910 Attn: Carol Earp	2
COMPATWINGSPAC Naval Air Station Moffett Field, CA 94035 Attn: Code 51 and Code 532	2
COMPATWINGSLANT Naval Air Station Brunswick, ME 04011 Attn: Code N7	2
Prof. R.H. Shudde, Code 55Su Naval Postgraduate School Monterey, CA 93940	100

DATE
ILME